# Computational Semantics, Type Theory, and Functional Programming 

II - Dynamic Montague Grammar and its Shortcomings

Jan van Eijck<br>CWI and ILLC, Amsterdam, Uil-OTS, Utrecht<br>LOLA7 Tutorial, Pecs

August 2002

## Summary

- Point of Departure: Dynamic Predicate Logic
- Extensions to Typed Logic
- Building a Montague Fragment from This
- States, Propositions, Transitions
- Implementation
- Chief Weakness


## Point of Departure: DPL

Dynamic Predicate Logic or DPL [GS91] is our point of departure.
Assume a first order model $M=(D, I)$ and a set of variables $V$. States for DPL are functions in $D^{V}$.

Let $M=(D, I)$ and let $s \in D^{V}$.
Term interpretation:
$\llbracket c \rrbracket_{s}^{M}=I(c)$ for constants $c$ and $\llbracket v \rrbracket_{s}^{M}=s(v)$ for variables $v$.

Define the relation

$$
M \models_{s} P t_{1} \cdots t_{n}
$$

by means of:

$$
M \models_{s} P t_{1} \cdots t_{n}: \Leftrightarrow\left\langle\llbracket t_{1} \rrbracket_{s}^{M}, \ldots, \llbracket t_{n} \rrbracket_{s}^{M}\right\rangle \in I(P)
$$

and the relation

$$
M \models_{s} t_{1} \doteq t_{2}
$$

by means of:

$$
M \models_{s} t_{1} \doteq t_{2}: \Leftrightarrow \llbracket t_{1} \rrbracket_{s}^{M}=\llbracket t_{2} \rrbracket_{s}^{M} .
$$

If $x \in V, d \in D$ and $s \in D^{V}$, use $(x \mid d) s$ for the state $s^{\prime}$ that differs from $s$ at most in the fact that $x$ gets mapped to $d$.
The DPL interpretation of formulas can now be given as a map in $D^{V} \rightarrow \mathcal{P}\left(D^{V}\right)$.

## DPL Semantics - Functional Version

$$
\begin{aligned}
& \llbracket \exists x \rrbracket(s):=\{(x \mid d) s \mid d \in D\} \\
& \llbracket P t_{1} \cdots t_{n} \rrbracket(s):= \begin{cases}\{s\} & \text { if } M \models_{s} P t_{1} \cdots t_{n} \\
\emptyset & \text { otherwise, }\end{cases} \\
& \llbracket t_{1} \doteq t_{2} \rrbracket(s):= \begin{cases}\{s\} & \text { if } M \models_{s} t_{1} \doteq t_{2} \\
\emptyset & \text { otherwise, }\end{cases} \\
& \llbracket \neg \varphi \rrbracket(s):= \begin{cases}\{s\} & \text { if } \llbracket \varphi \rrbracket(s)=\emptyset, \\
\emptyset & \text { otherwise, }\end{cases} \\
& \llbracket \varphi ; \psi \rrbracket(s):=\bigcup\left\{\llbracket \psi \rrbracket\left(s^{\prime}\right) \mid s^{\prime} \in \llbracket \varphi \rrbracket(s)\right\}
\end{aligned}
$$

Note that predicates, identities and negations are interpreted as tests: if $s$ is the input state they either return $\{s\}$, in case the test succeeds, or $\emptyset$, in case the test fails.

The action of existential quantification is not a test.
Dynamic implication $\varphi \Rightarrow \psi$ can be defined in terms of $\neg$ and ; by means of $\neg(\varphi ; \neg \psi)$. This is a test.
Universal quantification $\forall x \varphi$ is defined in terms of $\exists, \neg$ and $;$ as $\neg(\exists x ; \neg \varphi)$.
This is a test.

## DPL Semantics - Relational Version

$$
M, s, s^{\prime} \models \exists x: \Longleftrightarrow \text { there is a } d \in D \text { with } s^{\prime}=(x \mid d) s
$$

$$
\begin{gathered}
M, s, s^{\prime} \models P t_{1} \cdots t_{n}: \Longleftrightarrow s=s^{\prime} \text { and } M \models_{s} P t_{1} \cdots t_{n} \\
M, s, s^{\prime} \models t_{1} \doteq t_{2}: \Longleftrightarrow s=s^{\prime} \text { and } M \models_{s} t_{1} \doteq t_{2}
\end{gathered}
$$

$$
M, s, s^{\prime} \models \neg \varphi:=s=s^{\prime} \text { and there is no } s^{\prime \prime} \text { with } M, s, s^{\prime \prime} \models \varphi
$$

$M, s, s^{\prime} \models \varphi ; \psi:=$ there is an $s^{\prime \prime}$ with $M, s, s^{\prime \prime} \models \varphi$ and $M, s^{\prime \prime}, s^{\prime} \models \psi$.

## States as Carriers of Anaphoric Information

The advantage of the propagation of variable states is that they carry anaphoric information that can be used for the interpretation of subsequent discourse.

1 Some ${ }^{1}$ man loved some ${ }^{2}$ woman.
The DPL rendering of (1) is $\exists u_{1} ; M u_{1} ; \exists u_{2} ; W u_{2} ; L u_{1} u_{2}$.
This gets interpreted as the set of all maps $u_{1} \mapsto e_{1}, u_{2} \mapsto e_{2}$ that satisfy the relation 'love' in the model under consideration.
The result of this is that the subsequent sentence (2) can now use this discourse information to pick up the references:
$2 \mathrm{He}_{1}$ kissed her ${ }_{2}$.

## Extensions to Typed Logic

Attempts to incorporate DPL stype dynamic semantics in mainstream Montague style natural language semantics [Mon73] can be found in [GS90, Chi92, Jan98, Mus95, Mus96, Mus94, Eij97, EK97, KKP96, Kus00].
The type hierarchy employed has basic types for entities (e), truth values $(t)$, and markers $(m)$. The states themselves can be viewed as maps from markers to suitable referents, i.e., a state has type $m \rightarrow e$. We abbreviate this as $s$.
This can either be built into the type system from the start (see [Eij97]) or enforced by means of axioms (a kind of meaning postulates for proper state behaviour; see [Mus95, Mus94]).

Various set-ups of the encoding of DPL to type theory are possible. In the most straightforward approach, the meaning of a formula is no longer a truth value, but a state transition, i.e., the interpretations of formulas have type $s \rightarrow s \rightarrow t$.
Following [Eij97], we will take $(u \mid x)$ as a primitive operation of type $s \rightarrow s$ that resets the value of $u$ to $x$.
Thus, $(u \mid x) a$ denotes the state $a^{\prime}$ that differs from state $a$ at most in the fact that the value in $a^{\prime}$ for marker $u$ is (the interpretation of) $x$.
Then the translation of an indefinite noun phrase a man becomes something like:

$$
3 \lambda P \lambda a \lambda a^{\prime} . \exists x\left(\operatorname{man} x \wedge P u_{i}\left(u_{i} \mid x\right) a a^{\prime}\right)
$$

Here $P$ is a variable of type $m \rightarrow s \rightarrow s \rightarrow t$ and $a, a^{\prime}$ are variables of type $s$, so the translation (3) has type $(m \rightarrow s \rightarrow s \rightarrow t) \rightarrow s \rightarrow$ $s \rightarrow t$.

## State Transitions

Note that $s \rightarrow s \rightarrow t$ is the type of a (characteristic function of) a binary relation on states, or, as we will call it, the type of a state transition.

In (the present version of) dynamic semantics, VPs are interpreted as maps from markers to state transitions.

Translation (3) introduces an anaphoric index $i$; as long as $u_{i}$ does not get reset, any reference to $u_{i}$ will pick up the link to the indefinite man that was introduced into the discourse.

## Encodings of Dynamic Operations in Typed Logic

Assume $\varphi$ and $\psi$ have the type of state transitions, i.e., type $s \rightarrow s \rightarrow$ $t$, and that $a, a^{\prime}, a^{\prime \prime}$ have type $s$.

$$
\begin{gathered}
\exists u_{i}:=\lambda a a^{\prime} \cdot \exists x\left(\left(u_{i} \mid x\right) a=a^{\prime}\right) \\
\varpi \varphi:=\lambda a a^{\prime} \cdot\left(a=a^{\prime} \wedge \neg \exists a^{\prime \prime} \varphi a a^{\prime \prime}\right) \\
\varphi ; \psi:=\lambda a a^{\prime} \cdot \exists a^{\prime \prime}\left(\varphi a a^{\prime \prime} \wedge \psi a^{\prime \prime} a^{\prime}\right) \\
\varphi \Rightarrow \psi:=\sqcap(\varphi ; \sqcap \psi)
\end{gathered}
$$

It is also useful to define an operation ! : $(s \rightarrow t) \rightarrow t$ to indicate that the state set $s \rightarrow t$ is not empty. Thus, ! serves as an indication of success. Assume $p$ to be an expression of type $s \rightarrow t$, the definition of ! is:

$$
!p:=\exists a \cdot(p a) .
$$

Note that $\pi$ can now be defined in terms of!, as

$$
\rightarrow \varphi:=\lambda a a^{\prime} .\left(a=a^{\prime} \wedge \neg!\varphi a\right)
$$

## Lifting Lexical Meanings to the Discourse Level

We have to assume that the lexical meanings of CNs, VPs are given as one-placed predicates (type $e \rightarrow t$ ) and those of TVs as two-placed predicates (type $e \rightarrow e \rightarrow t$ ).
It makes sense to define blow-up operations for lifting one-placed and two-placed predicates to the dynamic level. Assume $A$ to be an expression of type $e \rightarrow t$, and $B$ an expression of type $e \rightarrow e \rightarrow t$ ); we use $r, r^{\prime}$ as variables of type $m, a, a^{\prime}$ as variables of type $s=m \rightarrow e$, and we employ postfix notation for the lifting operations:

$$
\begin{aligned}
& A^{\circ}:=\lambda r \lambda a \lambda a^{\prime}\left(a=a^{\prime} \wedge A(a r)\right) \\
& B^{\bullet}:=\lambda r \lambda r^{\prime} \lambda a \lambda a^{\prime}\left(a=a^{\prime} \wedge B(a r)\left(a r^{\prime}\right)\right)
\end{aligned}
$$

The encodings of the DPL operations in typed logic and the blowup operations for one- and two-placed predicates are employed in the semantic specification of the fragment. The semantic specifications employ variables $P, Q$ of type $m \rightarrow s \rightarrow s \rightarrow t$, variables $u, u^{\prime}$ of type $m$, and variables $a, a^{\prime}$ of type $s$.

| $\mathbf{S}$ | $::=$ NP VP | $X::=\left(X_{1} X_{2}\right)$ |
| :--- | :--- | :--- |
| S | $::=$ if S S | $X::=X_{2} \Rightarrow X_{3}$ |
| NP $::=$ Mary $^{n}$ | $X::=\lambda P a a^{\prime} .\left(P u_{n} a a^{\prime}\right)$ |  |
| NP $::=$ Bill $^{n}$ | $X::=\lambda P a a^{\prime} .\left(P u_{n} a a^{\prime}\right)$ |  |
| NP $::=$ PRO $^{n}$ | $X::=\lambda P a a^{\prime} .\left(P u_{n} a a^{\prime}\right)$ |  |
| NP $::=$ DET CN | $X::=\left(X_{1} X_{2}\right)$ |  |
| NP $::=$ DET RCN | $X::=\left(X_{1} X_{2}\right)$ |  |

$$
\begin{array}{ll}
\text { DET }: & :=\text { every }^{n} \\
X \quad::=\lambda P Q .\left(\exists u_{n} ; P u_{n}\right) \Rightarrow Q u_{n} \\
\text { DET }::=\text { some }^{n} \\
X \quad: \quad=\lambda P Q \cdot \exists u_{n} ; P u_{n} ; Q u_{n} \\
\text { DET }::=n o^{n} \\
X \quad::=\lambda P Q . \backsim\left(\exists u_{n} ; P u_{n} ; Q u_{n}\right) \\
\text { DET }::= & \text { the } \\
X \quad:= & \lambda P Q . \\
& \lambda a a^{\prime} . a=a^{\prime} \wedge \exists x \forall y\left(!\left(\exists u_{n} ; P u_{n}\left(u_{n} \mid y\right) a\right) \leftrightarrow x=y\right) ; \\
& \quad \exists u_{n} ; P u_{n} ; Q u_{n}
\end{array}
$$

CN ::= man
CN ::= woman
CN ::= boy
RCN ::= CN that VP
RCN ::= CN that NP TV
VP $::=$ laughed
VP ::= smiled
VP ::= TV NP
TV ::= loved
TV ::= respected
$X::=M^{\circ}$
$X::=W^{\circ}$
$X::=B^{\circ}$
$X::=\lambda u \cdot\left(\left(X_{1} u\right) ;\left(X_{3} u\right)\right)$
$X::=\lambda u \cdot\left(\left(X_{1} u\right) ;\left(X_{3}\left(\lambda u^{\prime} .\left(\left(X_{4} u^{\prime}\right) u\right)\right)\right)\right.$
$X::=L^{\circ}$
$X::=S^{\circ}$
$X::=\lambda u .\left(X_{2} ; \lambda u^{\prime} .\left(\left(X_{1} u^{\prime}\right) u\right)\right)$
$X::=L^{\prime \bullet}$
$X::=R^{\bullet}$

## Comparison with Classical Montague Grammar

The types of the dynamic meanings are systematically related to the types of the earlier static meanings by a replacement of truth values (type $t$ ) by transitions (type $s \rightarrow s \rightarrow t$ ), and of entities (type $e$ ) by markers (type $m$ ).
The translation of the proper names assumes that every name is linked to an anchored marker (a marker that is never updated).

## Implementation - Basic Types

module LOLA2 where<br>import Domain<br>import Model

Apart from Booleans and Entities, we need basic types for (reference) markers. Reference markers are the dynamic variables that carry discourse information. For convenience, we also declare a type for indices, and a map from indices to markers.

```
data Marker = U0 | U1 | U2 | U3 | U4
        | U5 | U6 | U7 | U8 | U9
    deriving (Eq,Bounded,Enum,Show)
type Idx = Int
i2m :: Idx -> Marker
i2m i | i < 0 || i > fromEnum (maxBound::Marker)
    = error "idx out of range"
    | otherwise
        = toEnum i
```


## States, Propositions

The type of states is Marker -> Entity.
For purposes of implementation, we will represent maps from markers to entities as lists of marker/entity pairs.

```
type State = [(Marker,Entity)]
```

Propositions are collections of states. We will represent propositions as lists of states:

$$
\text { type Prop }=\text { [State] }
$$

## Transitions

Transitions are mappings from states to propositions, i.e., their type is State -> Prop.
type Trans = State -> Prop

## Applying a state to a marker

Next, we need a function for application of a state to a marker. An error message is generated if the marker is not in the domain.

```
apply :: State -> Marker -> Entity
apply [] m = error (show m ++ " not in domain")
apply ((m,e):xs) m' | m == m' = e
    | otherwise = apply xs m'
```


## Updates

Updating a reference marker by mapping it to a new entity in a state (the implementation of the operation $(u \mid x)$ ):

```
update :: Marker -> Entity -> State -> State
update m e s = replace m e s where
    replace _ _ [] = error "undefined"
    replace m e ((m',e'):xs)
    | m == m' = (m,e):xs
    | otherwise = (m',e'):(replace m e xs)
```


## Dynamic Negation

Dynamic negation is a test on a state $s$ that succeeds of the negated formula fails in that state, and succeeds otherwise.

```
neg :: Trans -> Trans
neg = \ phi s -> if (phi s == []) then [s] else []
```


## Dynamic Conjunction

Dynamic conjunction applies the first conjunct to the initial state, next applies the second conjuncts to all intermediate results, and finally collects all end results.

```
conj :: Trans -> Trans -> Trans
conj = \ phi psi s ->
    concat [ psi s' | s' <- (phi s) ]
```


## Dynamic Quantification

Dynamic existential quantification is defined in terms of update, making use of the fact that the domain of entities is bounded.

```
exists :: Marker -> Trans
exists = \ m s ->
    [ update m x s | x <- entities]
```

The universal quantifier is defined in terms of dynamic existential quantification, dynamic conjunction and dynamic negation.

```
forall :: Marker -> Trans -> Trans
forall = \ m phi ->
    neg ((exists m) 'conj' (neg phi))
```


## Dynamic Implication

Dynamic implication is defined in terms of dynamic conjunction and dynamic negation.

```
impl :: Trans -> Trans -> Trans
impl = \ phi psi -> neg (phi 'conj' (neg psi))
```


## Anchors for Proper Names

To get a reasonable treatment of proper names, we assume that some of the discourse markers are anchored to entities:

```
anchored :: Marker -> Entity -> Bool
anchored U6 A = True
anchored U7 M = True
anchored U8 B = True
anchored U9 J = True
anchored _ _ = False
```


## Syntax of the Fragment

The datatype declarations for syntax are almost as for classical Montague grammar. The main difference is that all noun phrases now carry index information. The index on proper names and pronouns is directly attached to the name or pronoun; the index information on a complex NP is attached to the determiner.

```
data S = S NP VP | If S S | Txt S S
    deriving (Eq,Show)
data NP = Ann Idx | Mary Idx | Bill Idx
        | Johnny Idx | PRO Idx
        | NP1 DET CN | NP2 DET RCN
    deriving (Eq,Show)
data DET = Every Idx | Some Idx | No Idx | The Idx
    deriving (Eq,Show)
```

```
data CN = Man | Woman | Boy | Person | Thing | House
    deriving (Eq,Show)
data RCN = CN1 CN VP | CN2 CN NP TV
    deriving (Eq,Show)
data VP = Laughed | Smiled | VP TV NP
    deriving (Eq,Show)
data TV = Loved | Respected | Hated | Owned
    deriving (Eq,Show)
```


## Lexical Meaning

We intend to use the information from the first order model in module Model. Thus, we assume that lexical VPs and CNs have a lexical meaning of type Entity -> Bool, which is subsequently blown up to a suitable discourse type. Similarly for two-placed predicates.

Mapping one-place predicates to functions from markers to transitions (or: discourse predicates) is done by:

$$
\begin{aligned}
\text { blowupPred :: (Entity } & \text {-> Bool) -> Marker } \\
\text { blowupPred pred m s } & \text { | pred (apply s m) } \\
& =[s] \\
& \mid \text { otherwise }
\end{aligned}
$$

Discourse blow-up for two-place predicates.

$$
\begin{aligned}
& \text { blowupPred2 :: (Entity } \rightarrow \text { Entity } \rightarrow \text { Bool) -> } \\
& \text { Marker } \rightarrow \text { Marker } \rightarrow \text { Trans }
\end{aligned} \quad \begin{aligned}
\text { blowupPred2 pred } \mathrm{m} 1 \mathrm{~m} 2 \mathrm{~s}
\end{aligned} \quad \begin{aligned}
\text { | pred (apply s m1) (apply s m2) } & =[\mathrm{s}] \\
\text { | otherwise } & =[]
\end{aligned}
$$

## Dynamic Interpretation

The interpretation of sentences now produces transitions (type Trans) rather than booleans:

```
intS :: S -> Trans
intS (S np vp) = (intNP np) (intVP vp)
intS (If s1 s2) = (intS s1) 'impl' (intS s2)
intS (Txt s1 s2) = (intS s1) 'conj' (intS s2)
```

In fact, we can get at the types of all the translation instructions by systematically replacing Bool by Trans, and Entity by Marker.

Interpretation of proper names: the code checks whether the proper names employ a suitably anchored marker, and generates an error message in case they are not.

```
intNP :: NP -> (Marker -> Trans) -> Trans
intNP (Ann i) p | anchored (i2m i) A = p (i2m i)
    | otherwise = error "wrong anchor"
intNP (Mary i) p | anchored (i2m i) M = p (i2m i)
    | otherwise = error "wrong anchor"
intNP (Bill i) p | anchored (i2m i) B = p (i2m i)
    | otherwise = error "wrong anchor"
intNP (Johnny i) p | anchored (i2m i) J = p (i2m i)
    | otherwise = error "wrong anchor"
```

Interpretation of pronouns: use the anchor indicated by the index:

```
intNP (PRO i) p = p (i2m i)
```

Interpretation of complex NPs: use the appropriate recursion.

```
intNP (NP1 det cn) p = (intDET det) (intCN cn) p
intNP (NP2 det rcn) p = (intDET det) (intRCN rcn) p
```

Interpretation of lexical VPs uses the dynamic blow-up from the lexical meanings.

```
intVP :: VP -> Marker -> Trans
intVP Laughed subj = blowupPred laugh subj
intVP Smiled subj = blowupPred smile subj
```

Complex VPs:

```
intVP (VP tv np) subj = intNP np phi
    where phi obj = intTV tv obj subj
```

Interpretation of TVs uses discourse blow-up of two-place predicates.

```
intTV :: TV -> Marker -> Marker -> Trans
intTV tv = blowupPred2 (lexTV tv)
    where lexTV Loved = (flip . curry) love
    lexTV Respected = (flip . curry) respect
    lexTV Hated = (flip . curry) hate
    lexTV Owned = (flip . curry) own
```

Interpretation of CNs uses discourse blow-up of one-place predicates.

```
intCN :: CN -> Marker -> Trans
intCN Man = blowupPred man
intCN Boy = blowupPred boy
intCN Woman = blowupPred woman
intCN Person = blowupPred person
intCN Thing = blowupPred thing
intCN House = blowupPred house
```

Code for checking that a discourse predicate is unique.

```
unique :: Marker -> Trans -> Trans
unique m phi s | singleton xs = [s]
        | otherwise = []
    where singleton [x] = True
    singleton _ = False
    xs = [x | x <- entities,
    success (update m x s) phi]
    success s psi = phi s /= []
```

Discourse type of determiners: combine two discourse predicates into a transition.

```
intDET :: DET ->
    (Marker -> Trans) -> (Marker -> Trans) -> Trans
```

Interpretation of determiners in terms of the dynamic operators defined above.

```
intDET (Some i) phi psi =
    exists m 'conj' (phi m) 'conj' (psi m)
    where m = i2m i
intDET (Every i) phi psi =
    (exists m 'conj' (phi m)) 'impl' (psi m)
    where m = i2m i
intDET (No i) phi psi =
    neg (exists m 'conj' (phi m) 'conj' (psi m))
    where m = i2m i
intDET (The i) phi psi =
    (unique m (phi m)) 'conj'
    (exists m) 'conj' (phi m) 'conj' (psi m)
    where m = i2m i
```

Interpretation of relativised common nouns: straightforward generalisation of the treatment in classical Montague grammar:

```
intRCN :: RCN -> Marker -> Trans
intRCN (CN1 cn vp) m =
        conj (intCN cn m) (intVP vp m)
intRCN (CN2 cn np tv) m =
    conj (intCN cn m) (intNP np (intTV tv m))
```


## Testing It Out

We need a suitable start state. Note that this start states respects the anchoring information for the proper names.

```
startstate :: State
startstate =
    [(U0,A), (U1,B), (U2, C), (U3,D), (U4,E),
        (U5, F), (U6 , A), (U7 , M), (U8 , B), (U9 , J)]
```

Evaluation starts out from the initial state and produces a list of states. An empty output state set indicates that the sentence is false, a nonempty output state set indicates that the sentence is true. The output states encode the anaphoric discourse information that is available for subsequent discourse.

```
eval :: S -> Prop
eval s = intS s startstate
```

LOLA2> eval (S (Johnny 10) Smiled)

Program error: idx out of range

LOLA2> eval (S (Johnny 8) Smiled)

Program error: wrong anchor

LOLA2> eval (S (Johnny 9) Smiled)
[]

LOLA2> eval (S (Mary 7) Smiled)
[[(U0, A), (U1 , B) , (U2, C) , (U3, D) , (U4, E), (U5, F) , (U6 , A) , (U7 , M) , (U8 , B) , (U9, J)]]

```
test4 = eval
    (S (NP1 (The 1) Boy)
    (VP Loved (NP1 (Some 2) Woman)))
```

LOLA2> test4
[[(U0, A), (U1 , B) , (U2 , M) , (U3, D) , (U4, E), (U5, F), (U6, A), (U7, M), (U8 , B), (U9, J)]]

```
test5 = eval
    (S (NP1 (Some 1) Man)
    (VP Loved (NP1 (Some 2) Woman)))
```

LOLA2> test5
[[(U0, A), (U1 , B) , (U2 , M) , (U3, D), (U4, E), (U5, F), (U6, A), (U7, M), (U8, B), (U9, J)], $[(U 0, A),(U 1, J),(U 2, M),(U 3, D),(U 4, E)$, (U5, F), (U6, A), (U7, M), (U8 , B) , (U9, J)]]

```
test6 = eval
```

    (S (NP1 (Some 1) Man)
    (VP Respected (NP1 (Some 2) Woman)))
    LOLA2> test6
[ [ (U0, A) , (U1 , B) , (U2, A) , .] , $[(U 0, A),(U 1, B),(U 2, C), \ldots]$, $[(U 0, A),(U 1, B),(U 2, M), \ldots]$, $[(U 0, A),(U 1, D),(U 2, A), \ldots]$, $[(U 0, A),(U 1, D),(U 2, C), \ldots]$, $[(U 0, A),(U 1, D),(U 2, M), \ldots]$, $[(\mathrm{UO}, \mathrm{A}),(\mathrm{U} 1, \mathrm{~J}),(\mathrm{U} 2, \mathrm{~A}), \ldots]$, $[(U 0, A),(U 1, J),(U 2, C), \ldots]$,
$[(\mathrm{UO}, \mathrm{A}),(\mathrm{U} 1, \mathrm{~J}),(\mathrm{U} 2, \mathrm{M}), \ldots]]$

```
test7 = eval
    (S (NP1 (The 1) Man)
    (VP Loved (NP1 (Some 2) Woman)))
```

LOLA2> test7
[]

```
test8 = eval
    (S (NP1 (Some 1) Man)
    (VP Loved (NP1 (Some 2) Woman)))
```

LOLA2> test8
[[(U0, A), (U1 , B) , (U2 , M) , (U3, D) , (U4, E), (U5, F) , (U6, A), (U7, M), (U8, B), (U9, J)], [(U0, A), (U1, J), (U2, M), (U3, D), (U4, E), (U5,F), (U6, A), (U7, M), (U8, B), (U9, J)]]

```
test9 = eval
    (Txt (S (NP1 (Some 1) Man)
                (VP Loved (NP1 (Some 2) Woman)))
(S (PRO 1) (VP Respected (PRO 2))))
```

LOLA2> test9
[[(U0, A), (U1 , B) , (U2 , M) , (U3, D) , (U4 , E),
(U5, F) , (U6, A) , (U7, M) , (U8, B) , (U9, J)],
[(U0, A), (U1, J), (U2, M), (U3, D), (U4, E),
(U5,F), (U6, A), (U7, M), (U8, B), (U9, J)]]

```
test10 = eval
    (If (S (NP1 (Some 1) Man)
        (VP Loved (NP1 (Some 2) Woman)))
        (S (PRO 1) (VP Respected (PRO 2))))
```

LOLA2> test10
[[(U0, A), (U1 , B) , (U2 , C) , (U3, D) , (U4 , E) ,
(U5, F) , (U6, A), (U7 , M), (U8 , B) , (U9, J)]]

```
test11 = eval
    (Txt (S (NP1 (Some 1) Man)
                (VP Respected (NP1 (Some 2) Woman)))
(S (PRO 1) (VP Loved (PRO 2))))
```

LOLA2> test11
[[(U0, A), (U1 , B) , (U2 , M) , (U3, D), (U4, E),
(U5, F) , (U6, A), (U7, M) , (U8, B) , (U9, J)],
[(U0, A), (U1, J), (U2, M), (U3, D), (U4, E),
(U5, F) , (U6 , A), (U7, M), (U8 , B) , (U9, J)]]

```
test12 = eval
    (If (S (NP1 (Some 1) Man)
        (VP Respected (NP1 (Some 2) Woman)))
    (S (PRO 1) (VP Loved (PRO 2))))
```

LOLA2> test12
[]

## Chief Weakness

The chief weakness of DPL-based NL semantics is the need to supply indices for all noun phrases in advance.
Moreover, re-use of an index destroys access to the previous value of the corresponding marker.

This is a reflection of the fact that DPL has destructive assignment: quantification over $u$ replaces the previous contents of register $u$ by something new, and the old value gets lost forever.

## References

[Chi92] G. Chierchia. Anaphora and dynamic binding. Linguistics and Philosophy, 15(2):111-183, 1992.
[Eij97] J. van Eijck. Typed logics with states. Logic Journal of the IGPL, 5(5):623-645, 1997.
[EK97] J. van Eijck and H. Kamp. Representing discourse in context. In J. van Benthem and A. ter Meulen, editors, Handbook of Logic and Language, pages 179-237. Elsevier, Amsterdam, 1997.
[GS90] J. Groenendijk and M. Stokhof. Dynamic Montague Grammar. In L. Kalman and L. Polos, editors, Papers from the Second Symposium on Logic and Language, pages 3-48. Akademiai Kiadoo, Budapest, 1990.
[GS91] J. Groenendijk and M. Stokhof. Dynamic predicate logic. Linguistics and Philosophy, 14:39-100, 1991.
[Jan98] Martin Jansche. Dynamic Montague Grammar lite. Dept of Linguistics, Ohio State University, November 1998.
[KKP96] M. Kohlhase, S. Kuschert, and M. Pinkal. A type-theoretic semantics for $\lambda$-DRT. In P. Dekker and M. Stokhof, editors, Proceedings of the Tenth Amsterdam Colloquium, Amsterdam, 1996. ILLC.
[Kus00] S. Kuschert. Dynamic Meaning and Accommodation. PhD thesis, Universität des Saarlandes, 2000. Thesis defended in 1999.
[Mon73] R. Montague. The proper treatment of quantification in ordinary English. In J. Hintikka e.a., editor, Approaches to Natural Language, pages 221-242. Reidel, 1973.
[Mus94] R. Muskens. A compositional discourse representation theory. In P. Dekker and M. Stokhof, editors, Proceedings 9th Amsterdam Colloquium, pages 467-486. ILLC, Amsterdam, 1994.
[Mus95] R. Muskens. Tense and the logic of change. In U. Egli et al., editor, Lexical Knowledge in the Organization of Language, pages 147-183. W. Benjamins, 1995.
[Mus96] R. Muskens. Combining Montague Semantics and Discourse Representation. Linguistics and Philosophy, 19:143-186, 1996.

